Model H1:

```
model{
for(i in 1:N){
        p[i,1] <- 1
        for (j in 1:nc[i]-1) {
                r[i,j] ~ dbin(q[i,j],n[i,j])
                q[i,j] <- 1-(p[i,C[i,j+1]]/p[i,C[i,j]])
                z.index[i,j]<- C[i,j+1]-1
                theta[i,j] <- mu[s[i]] + (d[t[i]] - d[t[1]])*(1-equals(t[i],b[i])) + z[z.index[i,j]]
                rhat[i,j] <- q[i,j] * n[i,j]
                dv[i,j] <- 2 * (r[i,j]*(log(r[i,j])-log(rhat[i,j])) + (n[i,j]-r[i,j])*(log(n[i,j]-r[i,j]) - log(n[i,j]-
        rhat[i,j])))
                        }
                dev[i] <- sum(dv[i,1:nc[i]-1])
        for (j in 2:nc[i]) {
                p[i,C[i,j]] <- 1 - phi.adj[i,j]
                phi.adj[i,j] <- phi(theta[i,j-1])
                        }
        }
totresdev <- sum(dev[])
z[1] <- 0
for (j in 2:Cmax-1) {
        z.aux[j] ~ dunif(0,5)
        z[j] <- z[j-1] + z.aux[j]
                }
d[1] <- 0
for (k in 2:nt){ d[k] ~ dnorm(0,0.000001) }
for(i in 1:ns){ mu[i] ~ dnorm(0,0.000001)}

for (i in 1:ns) {
mu1[i]<-mu[i]*equals(t[1],1)}
A<-sum(mu1[])/ns

# calculate prob of achieving ACR20/50/70 on treat k
for (k in 1:nt) {
for (j in 1: Cmax-1) { T[j,k] <- 1 - phi(A + d[k] + z[j]) }
}
}
```

Model K2:

```
model{
for(i in 1:N){
        p[i,1] <- 1
        for (j in 1:nc[i]-1) {
                r[i,j] ~ dbin(q[i,j],n[i,j])
                q[i,j] <- 1-(p[i,C[i,j+1]]/p[i,C[i,j]])
                z.index[i,j]<- C[i,j+1]-1
                theta[i,j] <- mu[s[i]] + d[t[i]] + z[z.index[i,j]]
                        + betaplac * (mu[s[i]] - Mean) * (1-equals(t[i],1))
                rhat[i,j] <- q[i,j] * n[i,j]
                dv[i,j] <- 2 * (r[i,j]*(log(r[i,j])-log(rhat[i,j])) + (n[i,j]-r[i,j])*(log(n[i,j]-r[i,j]) - log(n[i,j]-rhat[i,j])))
                        }
                dev[i] <- sum(dv[i,1:nc[i]-1])
        for (j in 2:nc[i]) {
          p[i,C[i,j]] <- 1 - phi.adj[i,j]
```

```
            phi.adj[i,j] <- phi(theta[i,j-1])
          }
}
totresdev <- sum(dev[])
z[1] <- 0

for (j in 2:Cmax-1) {
          z.aux[j] ~ dunif(0,5)
          z[j] <- z[j-1] + z.aux[j]
                              }
d[1] <- 0
for (k in 2:4){ d[k] ~dnorm( D.c[1], prec.d) }
for (k in 5:9){ d[k] ~dnorm( D.c[2], prec.d) }
d[10] <-D.c[3]
for (i in 1:3) {D.c[i] ~ dnorm(0,0.01) }
prec.d<- 1/(sd.d*sd.d)
sd.d~dunif(0,10)
for (i in 1:2) {D.pred[i]~dnorm(D.c[i],prec.d)}
for(i in 1:ns){ mu[i] ~ dnorm(0,0.01)}
betaplac ~ dnorm(0,0.01)

for (i in 1:ns) {
mu1[i]<-mu[i]*equals(t[1],1)}
A<-sum(mu1[])/ns

# calculate prob of achieving ACR20/50/70 on treat k
for (k in 1:nt) {
for (j in 1: Cmax-1) { T[j,k] <- 1 - phi(A + d[k] + z[j]) }
}
}
```

d[1]=PLA, d[2]=SEC300, d[3]=SEC150, d[4]=UST, d[5]=CZP, d[6]=GOL, d[7]=ADA, d[8]=INF, d[9]=ETA, d[10]=APR